
QU I C K  R E F E R E N C E  ·  TK70

Linear Hall Sensor
How strong a magnetic field is, and which pole: an analog voltage around half of VCC.

VCC is your board's logic voltage

The output scales with VCC and reaches about 0.8 of it in a strong
field. On an ESP32, ESP32-S3 or Pico wire VCC to 3V3: from 5 V a
magnet near the chip takes SIGNAL to about 4 V, over what a 3.3 V
pin takes. Only an Uno takes 5V.

Pins, and which to wire
Parts up, header at the bottom. Wire GND, VCC and SIGNAL to an analog pin; NC is connected to nothing
on the board. There is no LED.

1 2 3 4

Front GND VCC NC SIGNAL

Read it GND VCC SIGNAL

SLSS49E-3 Hall chip
Hold a magnet's face
over it, not edge-on.

100 nF across VCC
Steadies the chip's
supply. No LED here.

GND is the square pad
Count from it. The back
prints no pin names.

North at the front raises it
By the sheet's drawing.

Behind the board: swapped.

100 nF on SIGNAL
A filter far faster
than any hand.

NC
Connected to nothing

on the board.

No field reads half scale, not 0 SIGNAL rests at half of VCC, about 512 on an Uno. A field moves it up for one pole
and down for the other. Chips differ by up to ±77 G at rest, so measure the zero at
start-up, with no magnet near.

What a gauss is worth, by supply

No field

½ of VCC

2.25 to 2.75 V at 5 V, about 1.65 at 3.3 V.

VCC on 5V

3.25 mV per
gauss

2.95 to 3.55. Linear to ±460 G, 1 to 4 V.

VCC on 3V3

2.1 mV per
gauss, about

Read off the sheet's typical curve. The
output scales with VCC.

One count is about 1.5 G on an Uno or a Pico and about 0.36 G on an ESP32 on paper, where the ADC jitters by several counts.
Average readings. A wire carrying 5 A makes about 1 G at 1 cm: this is not a current sensor on its own.

Figures from Slkor's SLSS49E-3 data sheet at 25 °C: 3 to 12 V supply, 5 mA typical at 5 V, 3 µs response. Its drawings of which pole raises the output differ between
packages; check yours with a compass. The sketch compiles for an ESP32-S3 and an Uno; the block has not been bench-tested for this sheet.

The first reading, on an ESP32-S3
Arduino IDE · Board: ESP32S3 Dev Module · USB CDC On Boot: Enabled · No library

// The analog pin SIGNAL is wired to.

// Uno: A0. ESP32: 34. ESP32-S3: 4. Pico: 26.

const int HALL_PIN = 4;

const float MV_PER_GAUSS = 2.1;  // VCC on 3V3; 3.25 on 5V

const bool NORTH_RAISES = true;   // the sheet's SOT-23 drawing

float zeroMv;

float averageMv(int n) {

  float sum = 0;

  for (int i = 0; i < n; i++) {

    sum += analogReadMilliVolts(HALL_PIN);

  }

  return sum / n;

}

void setup() {

  Serial.begin(115200);

  delay(500);

  zeroMv = averageMv(64);   // no magnet near, please

}

void loop() {

  float gauss = (averageMv(16) - zeroMv) / MV_PER_GAUSS;

  Serial.print(gauss, 0);

  Serial.print(" G");

  if (fabs(gauss) >= 5) {

    bool north = (gauss > 0) == NORTH_RAISES;

    Serial.print(north ? "  north" : "  south");

  }

  Serial.println();

  delay(250);

}

0 G

212 G north

Serial Monitor at 115200, four lines a second: near 0 G with nothing near, a signed
field with the pole's name as a magnet comes to the chip.

Wiring
FRONT S3 UNO ESP32 PICO

1 GND GND GND GND GND

2 VCC 3V3 5V 3V3 3V3

3 NC

4 SIGNAL GPIO 4 A0 GPIO 34 GP26

Hole 1 is the square pad. VCC is each board's own logic
voltage: 3V3 on the ESP32 boards and the Pico, 5V on the
Uno. SIGNAL goes to an analog pin, ADC1 on the ESP32s.

Naming the pole, without flicker
A pole finder names north or south once the field passes 20 G and lets go only below 10 G, so a magnet held at the edge does not flicker. A magnet
behind the board swaps the names. Past ±460 G the reading says strong, not how strong.

When it does not work
Near 0 all the time
No power, or SIGNAL on another pin or on NC. Count from the square
pad.

Stuck near half scale
Working, no field on the chip. Hold a magnet's face over it.

Full scale with no magnet
ESP32 in MicroPython with no atten line: set ATTN_11DB.

A field from nothing
Zero taken with a magnet or steel near. Clear it and press reset.

North and south swapped
Magnet behind the board, or flip NORTH_RAISES after a compass
test.

Jumps by a few gauss
ESP32 ADC jitter. Average more readings; stay on ADC1.

Specifications
Chip SLSS49E-3, SS49E-type, SOT-23 No field Half of VCC; 2.25 to 2.75 V at 5 V

Sensitivity 3.25 mV/G at 5 V; about 2.1 at 3.3 V Linear range ±460 G: 1 V to 4 V at 5 V

Response 3 µs typical Supply 3 to 12 V; use your logic voltage

Current 3 to 8 mA at 5 V, always on SIGNAL Chip output, 100 nF to GND

